
OH HAI

.WHO

Perl 6 Hacker
Work on the Rakudo Perl 6 

compiler

Focus on the object model, type 

system and multiple dispatch

Traveller
Latvia is the 33rd country I have 

visited so far

Beer Drinker
Beer is tasty.

I drink it. 

☺☺☺☺

Perl 6
and

Rakudo

Perl 6

� Takes the key concepts of Perl…

� Influence of linguistics

� Get the job done

� Create a new, Perl-ish language with many 

similarities to Perl 5, but also many 

differences, e.g.

� Syntactic improvements (cleaner, better 

huffmanized)

� New features

Rakudo

� The most actively developed Perl 6 compiler

� Implements a wide range of Perl 6 features

� Currently only targets Parrot; in the future, 

we expect to support more backends (even 

within the coming months)

� We're reaching the point where they'll help 

rather than distract us from getting to a full 

Perl 6 specification and implementation

� Appeal to more users and developers

Rakudo Star

� A significant release of Rakudo scheduled 

for Spring 2010

� Not all of Perl 6, but a significant part of it

� A lot of the Perl 6 promised features can 

be found within Rakudo

� All code that I will show today runs on 

Rakudo already

� Aim to be usable and useful

What's this
talk about?

Overview

� I'll take a range of everyday programming 

problems and for each one show…

� The Perl 6 code that solves it

� The output that code gives when run

� Hopefully, a good way for you to start to 

grasp some of the new syntax and features

� A chance to show off some of the cool stuff 

in Perl 6

� All examples shown today work in Rakudo

say "Hello, world!"

Problem
Say "Hello, world"

Solution

Output
Hello, world!

print "Enter your name: ";

my $name = $*IN.get;

say "Hi $name!";

Problem
Read input from the console

Solution

Output
Enter your name: Jonathan

Hi Jonathan!



Solved in Perl 6

loop {

print "Enter a number from 1 to 10: ";

my $num = $*IN.get;

unless 1 <= $num <= 10 { say "Fail!" }

}

Problem
Check a value is in a given range

Solution 1

Output
Enter a number between 1 and 10: 3

Enter a number between 1 and 10: 42

Fail!

loop {

print "Enter a number from 1 to 10: ";

my $num = $*IN.get;

unless $num ~~ 1..10 { say "Fail!" }

}

Problem
Check a value is in a given range

Solution 2

Output
Enter a number between 1 and 10: 3

Enter a number between 1 and 10: 42

Fail!

my @nums = 1, 5, 7, -2, 3, 9, 11, -6, 14;

say [+] @nums;

Problem
Add up a list of numbers

Solution

Output
42

my @a = 1, 1, 2, 3, 5, 8;

my @b = 9, 4, 1, 16, 36, 25;

if [<=] @a { say '@a is sorted' }

if [<=] @b { say '@b is sorted' }

Problem
Check if a list is sorted

Solution

Output
@a is sorted

my @a = 1, 2, 3;

push @a, { x => 42, y => 100 };

say @a.perl;

Problem
Get a Perl-ish representation of a data structure 
(Data::Dumper style)

Solution

Output
[1, 2, 3, {"y" => 100, "x" => 42}]

my @cities = <Lisbon Tokyo Seoul Riga>;

for @cities -> $city {

say "I've been to $city";

}

Problem
Iterate over a list

Solution

Output
I've been to Lisbon

I've been to Tokyo

I've been to Seoul

I've been to Riga

my %distances =

Bratislava => 1084,

Stockholm  => 442; 

for %distances.kv -> $city, $distance {

say "$city is $distance km away";

}

Problem
Iterate over the keys and values of a hash

Solution

Output
Bratislava is 1084 km away

Stockholm is 442 km away

my @a = 75, 47, 90, 22, 80;

my @b = 61, 77, 94, 82, 60;

my @c = 45, 59, 33, 11, 19; 

if any(@a) >= 60 { say "Some passes in A" }

if any(@b) >= 60 { say "Some passes in B" }

if any(@c) >= 60 { say "Some passes in C" }

Problem
Check if any of a list of test scores is a pass

Solution

Output
Some passes in A

Some passes in B

my @a = 75, 47, 90, 22, 80;

my @b = 61, 77, 94, 82, 60;

my @c = 45, 59, 33, 11, 19; 

if all(@a) >= 60 { say "All passes in A" }

if all(@b) >= 60 { say "All passes in B" }

if all(@c) >= 60 { say "All passes in C" }

Problem
Check if all of a list of test scores are passes

Solution

Output
All passes in B

my @a = 75, 47, 90, 22, 80;

my @b = 61, 77, 94, 82, 60;

my @c = 45, 59, 33, 11, 19; 

if none(@a) >= 60 { say "No passes in A" }

if none(@b) >= 60 { say "No passes in B" }

if none(@c) >= 60 { say "No passes in C" }

Problem
Check if none of a list of test scores is a pass

Solution

Output
No passes in C

my @drinks = <wine beer vodka>;

say "Tonight I'll drink { @drinks.pick }";

Problem
Get a random item from a list

Solution

Output (results should vary ;-))
Tonight I'll drink vodka 

my @competitors = <Tina Lena Owen Peter>;

my @order = @competitors.pick(*);

for @order { .say }

Problem
Shuffle a list into a random order

Solution

Output (results should vary ;-))
Peter

Lena

Owen

Tina

sub greet($name) {

say "OH HAI $name!";

}

greet("masak");

Problem
Write and call a subroutine with parameters

Solution

Output
OH HAI masak

sub example(@a, %h) {

say @a.elems;

say %h.keys;

}

my @nums = 42, 57, 74;

my %mapping = a => 1, b => 2;

example(@nums, %mapping);

Problem
Write a subroutine taking an array and a hash

Solution

Output
3

ab

sub double(Num $n) { 2 * $n }

say double(21);

say double("oh no I'm not a number");

Problem
Write a subroutine that only takes a number

Solution

Output
42

Parameter type check failed; expected Num, 

but got Str for $n in call to double

multi double(Num $n) { 2 * $n }

multi double(Str $s) { $s x 2 }

say double(21);

say double("boo");

Problem
Use multi-subs to react differently by type

Solution

Output
42

booboo

multi fact($n) { $n * fact($n - 1) }

multi fact(0)  { 1 }

say fact(1);

say fact(10);

Problem
Compute factorial (recursively)

Solution

Output
1

3628800

sub fact($n) { [*] 1..$n }

say fact(1);

say fact(10);

Problem
Compute factorial (using a meta-operator)

Solution

Output
1

3628800

sub postfix:<!>($n) { [*] 1..$n }

say 1!;

say 10!;

Problem
Add a new factorial operator (so 10! works)

Solution

Output
1

3628800

class Product {

has $.name;  # Attr + accessor

has $!price; # Attr only

has $.discount is rw;

# Attr + lvalue accessor  

method get_price {

return $!price - $!discount;

}

}

Problem
Declare a class with attributes and a method

Solution

my $prod = Product.new(

name      => "Beer",

price     => 500,

discount  => 60

);

say $prod.get_price;

Problem
Instantiate a class and call a method on it

Solution

Output
440

say $prod.name;

$prod.discount = 40;

say $prod.get_price;

$prod.name = 'Wine';

Problem
Get/set attributes through accessors

Solution

Output
Beer

460

Cannot assign to readonly variable.

my @products =

Product.new(name => 'Beer', price => 500),

Product.new(name => 'Wine', price => 450),

Product.new(name => 'Vodka', price => 1600);

my @uc_names = @products>>.name>>.uc;

for @uc_names { .say }

Problem
Call a method on every object in a list

Solution

Output
BEER

WINE

VODKA

my @meths = Product.^methods(:local);

for @meths>>.name { .say }

Problem
Introspect a class to find its methods

Solution

Output
get_price

discount

name

my @products =

Product.new(name => 'Beer', price => 500),

Product.new(name => 'Wine', price => 450),

Product.new(name => 'Vodka', price => 1600);

my @sorted = @products.sort(*.name);

for @sorted { .name.say }

Problem
Sort an array of objects by result of a method

Solution (Example 1)

Output (Example 1)
Beer

Vodka

Wine

my @products =

Product.new(name => 'Beer', price => 500),

Product.new(name => 'Wine', price => 450),

Product.new(name => 'Vodka', price => 1600);

my @sorted = @products.sort(*.get_price);

for @sorted { .name.say }

Problem
Sort an array of objects by result of a method

Solution (Example 2)

Output (Example 2)
Wine

Beer

Vodka

my @temperatures = -3, 5, 7, 2, -1, -4, 0;

say "Minimum was " ~ @temperatures.min;

say "Maximum was " ~ @temperatures.max;

Problem
Find minimum and maximum values from a list

Solution (Example 1)

Output (Example 1)
Minimum was –4

Maximum was 7

my @products =

Product.new(name => 'Beer', price => 500),

Product.new(name => 'Wine', price => 450),

Product.new(name => 'Vodka', price => 1600);

say "Cheapest: " ~ @products.min(*.get_price).name;

say "Costliest: " ~ @products.max(*.get_price).name;

Problem
Find minimum and maximum values from a list

Solution (Example 2)

Output (Example 2)
Cheapest: Wine

Costliest: Vodka

class Paper { }

class Scissor { }

class Stone { }

multi win(Paper,   Stone)   { "Win" }

multi win(Scissor, Paper)   { "Win" }

multi win(Stone,   Scissor) { "Win" }

multi win(::T,     T)       { "Draw" }

multi win(Any,     Any)     { "Lose" }

Problem
Paper, Scissor, Stone game

Solution (Part 1)

say win(Paper, Paper);

say win(Scissor, Stone);

say win(Stone, Scissor);

Problem
Paper, Scissor, Stone game

Solution (Part 2)

Output
Draw

Lose

Win

Want to play with Perl 6?

� Rakudo – the most actively developed Perl 6 

compiler – is available from:

http://www.rakudo.org/

� Lots of Perl 6 resources can be found at:

http://www.perl6.org/

� Join the friendly IRC channel:

#perl6 on irc.freenode.org

� Write modules, write applications, jump into 

the evolving Perl 6 community and make 

your mark on it ☺



Thank You 

Questions?

